TEMPLATE = app

CONFIG(debug, debug|release) {
    TARGET = dokerd
} else {
    TARGET = doker
}

ROOTDIR = ..
DESTDIR = $$ROOTDIR/bin

QT += \
    core \
    gui \
    script

SOURCES += \
    appsettings.cpp \
    dokermodel.cpp \
    main.cpp \
    mainwindow.cpp \
    widgets/dokereditor.cpp \
    widgets/dokerwindow.cpp \
    widgets/sourcetree.cpp \
    widgets/sourcedock.cpp \
    widgets/logdock.cpp \
    dokerprocessor.cpp \
    objects/pageobject.cpp \
    objects/textobject.cpp

HEADERS += \
    appsettings.h \
    dokermodel.h \
    dokerversion.h \
    mainwindow.h \
    widgets/dokereditor.h \
    widgets/dokerwindow.h \
    widgets/sourcetree.h \
    widgets/sourcedock.h \
    qzip/qzipreader_p.h \
    qzip/qzipwriter_p.h \
    widgets/logdock.h \
    dokerprocessor.h \
    objects/pageobject.h \
    objects/textobject.h \
    doker.h

win32 {
RC_FILE += doker.rc
QMAKE_LFLAGS += -static-libgcc
}
RESOURCES = doker.qrc

TRANSLATIONS += \
    translations/doker_ru.ts
CODECFORTR = UTF-8

# Pre build settings
# Create our custom translate target.
translate.commands = $$QMAKE_LIBDIR_QT/../bin/lrelease translations/doker_ru.ts
QMAKE_EXTRA_TARGETS += translate
# Hook our translate target in between qmake's Makefile update and the actual project target.
translatehook.depends = translate
CONFIG(debug,debug|release):translatehook.target = Makefile.Debug
CONFIG(release,debug|release):translatehook.target = Makefile.Release
QMAKE_EXTRA_TARGETS += translatehook
